
Procesos por lotes

1. -Algunas cuestiones sobre procesos por lotes

1.1. - ¿Qué es un proceso por lotes?
Se podría definir, en una primera aproximación, como un conjunto de comandos que se ejecutan de 
forma  secuencial  para  realizar  alguna  tarea  concreta.  Esta  tarea  requiere  la  ejecución  de  estos 
comandos, de forma que se agrupan en un proceso por lotes para que sean ejecutados todos de forma 
secuencial, uno detrás de otro, en un determinado orden.

1.2. - ¿Qué ventajas proporciona?
Tener una serie de comandos agrupados en un proceso por lotes facilita la ejecución repetida de dicho 
conjunto de comandos y su reutilización de manera sencilla. 

Al crear un proceso por lotes, basta con escribir solo una vez el conjunto de comandos a ejecutar. A 
partir de ese momento, será suficiente con ejecutar el proceso por lotes en lugar de tener que volver a 
escribir y ejecutar todos los comandos (no tenemos que escribir una y otra vez estos comandos para 
realizar una tarea). 

Además, la ejecución de un conjunto de comandos secuenciales, obliga a que el usuario escriba 
estos comandos en orden uno a uno, teniendo que esperar a que termine un comando para poder 
ejecutar el siguiente.

1.3. - ¿De donde vienen?
Este concepto de proceso por lotes viene de la forma en que se ejecutaban una serie de instrucciones o 
comandos  secuenciales  en  las  primeras  generaciones  de  computadores.  Se  proporcionaban  los 
comandos a ejecutar en orden a través de tarjetas perforadas o cintas magnéticas, estos comandos se 
cargaban en el ordenador y se ejecutaban uno tras otro sin intervención del usuario, realizando una 
tarea concreta.

Los procesos por lotes en MS DOS y Windows, son algo similar, ya que están compuestos por una 
serie de comandos que se ejecutarán en orden (secuencialmente) para realizar una determinada tarea.

1.4. - ¿Cómo se implementa?
En MS DOS / Windows un proceso por lotes es  un archivo de texto plano, con extensión BAT o con 
extensión CMD (Windows XP), que contiene en cada una de sus líneas un comando. Estos comandos 
deben estar ordenados en el mismo orden en el que se quieran ejecutar.

Caso práctico

Crea un proceso por lotes, que muestre en pantalla la fecha y después la hora del sistema.  
No se desea que los comandos esperen por el cambio de fecha u hora, respectivamente. 

Lo primero que habría que hacer es analizar el problema y comprender que es lo que se nos 
pide.

Una  vez  comprendido  el  problema,  tratamos  de  solucionarlo.  Para  ello,  buscamos  los 
comandos en secuencia que permiten realizar la tarea que se nos pide. Lo primero es mostrar 
la fecha, para ello se utiliza el comando DATE. Después hay que mostrar la hora, para ello se 
utiliza el comando TIME. 

En este caso, se necesitan dos comandos: DATE y TIME. Además, deberíamos comprobar 
que se cumple la última condición del enunciado: “no se desea que los comandos esperen por 
el cambio de fecha u hora, respectivamente.”. Para ello debemos utilizar en los comandos el 
parámetro /T.

Javier Pedrero Martín
sif.javier  @gmail.com  

2



Sistemas Informáticos Multiusuario y en Red

Una vez solucionado el problema, lo implementamos, es decir, creamos el proceso por lotes. 
Creamos  un  archivo  de  texto  plano  y,  si  hace  falta,  le  damos  un  nombre.  Por  ejemplo, 
“fecha.BAT” (ojo es obligatorio que el fichero tenga extensión BAT).

Para ejecutar un proceso por lotes basta con escribir su nombre (no es necesario escribir su 
extensión) y pulsar Enter. Los procesos por lotes son tomados por el interprete como archivos 
ejecutables, es decir, los buscará en el directorio actual, y si no están ahí, en los directorios del 
PATH.

1.5. - ¿Qué herramienta utilizar?
Se puede utilizar  cualquier  herramienta que permita crear ficheros de texto plano y guardarlos con 
extensión  BAT  o  CMD.  Esto  es  lo  más  importante,  que  la  extensión  del  fichero  sea  una  de  las 
permitidas, para que el intérprete de comandos lo reconozca como un proceso por lotes.

1.6. - ¿Como se ejecutan los procesos por lotes?
La ejecución de un proceso por lotes se trata de una ejecución interpretada. Es decir, el intérprete de 
comandos, cuando se ejecuta un proceso por lotes, abre el fichero que lo implementa, y va leyendo y 
ejecutando comando a comando, hasta que termina (llega al final del fichero).

1.7. - ¿Qué comandos se pueden utilizar?
Se pueden ejecutar  cualquier  comando o  aplicación que se pueda ejecutar  desde el  intérprete  de 
comandos. Como se ha dicho, realmente lo que hace el intérprete de comandos es tomar el fichero BAT 
e interpretar o ejecutar cada una de sus líneas en orden.

De ahí que los procesos por lotes también se denominen scripts1.

1.7.1. - ¿Cuándo termina un proceso por lotes?
Un proceso por lotes termina cuando el intérprete llega al final del fichero o cuando se pulsa CTRL+C.

2. -Comandos de procesos por lotes
En un proceso por lotes se puede utilizar cualquier comando, interno o externo, o cualquier aplicación 
accesible desde el intérprete de comandos. Sin embargo, existe un subconjunto de comandos, cuyo uso 
está especialmente dirigido a su uso en procesos por lotes o ficheros BAT. A continuación vamos se 
verán algunos de ellos.

2.1. - Comando ECHO
El comando Echo sirve para mostrar un mensaje en pantalla y para activar o desactivar el eco de los 
comandos.

2.1.1. - Eco de los comandos
El eco de los comandos puede estar activo o desactivo. Si el eco2 está activo, entonces el intérprete 

de comandos mostrará en la pantalla cada comando que ejecuta. Si  no está activo, no mostrará el 
comando que se está ejecutando. Se suele desactivar el eco de pantalla para no “ensuciar” la pantalla 
con la secuencia de comandos que se ejecuta en un proceso por lotes.

ECHO OFF

DIR *.C /S

1 La palabra script  significa guión. Realmente este tipo de archivos lo único que proporcionan es un guión de 
ejecución de comandos. Los scripts también se utilizan en otros sistemas operativos, como Unix o Linux donde 
reciben el nombre de shell scripts. Su uso no está solo limitado a SS.OO. sino que también se utilizan para agilizar la 
ejecución de un conjunto de comandos secuenciales en aplicaciones como Photoshop o para crear programas como 
en JavaScript o VBScript.
2 La denominación de eco proviene de que el interprete lee y repite en pantalla el comando que va a ejecutar para 
que el usuario sea consciente de lo que está haciendo.

Código 1 – Ejemplo de desactivación del eco de comando.
El ejemplo anterior,  desactivaría el  eco de los comandos a ejecutar,  con lo que el  interprete de 

comandos no mostrará los comandos que ejecute. Acto seguido, el intérprete ejecutará el comando DIR 
*.C  /S,  pero  no  mostrará  en  pantalla  dicho  comando.  Pero,  atención,  si  mostrará  por  pantalla  el 
resultado de este comando.

Dos elementos a resaltar, primero el comando ECHO OFF si es mostrado por el intérprete. Esto es 
debido a que ese comando se ejecuta antes de que se desactive el eco. El segundo elemento a resaltar, 
es que el eco se vuelve a activar de forma automática cuando se termina la ejecución del proceso por 
lotes.

2.1.2. - Mostrar mensajes
El comando ECHO también nos permite mostrar mensajes al usuario. Para ello basta con escribir el 
comando, seguido de la cadena de texto que pretendamos mostrar, separada de un espacio.

ECHO OFF

ECHO Fecha y hora actual del sistema

DATE /T

TIME /T

Código 2 – Ejemplo para mostrar mensajes.

2.1.3. - Mostrar una línea vacía
Hay  veces  que  interesa  separar  líneas  en  la  pantalla,  por  ejemplo  para  separar  dos  mensajes 
mostrados al usuario. Para ello se utiliza el comando ECHO seguido por un punto.

ECHO OFF

ECHO Bienvenido al sistema.

ECHO.

ECHO La fecha y hora actuales son

DATE /T

TIME /T

Código 3 – Ejemplo para mostrar líneas en blanco.

2.1.4. - @
Como se ha visto, se puede desactivar el eco de los comandos para que el intérprete no muestre los 
comandos a medida que los ejecuta. Sin embargo, siempre se muestra el comando ECHO OFF, puesto 
que antes de ejecutarse el eco estaba activo.

Existe  otro  comando  que  permite  desactivar  el  eco  solo  con  el  comando  que  se  escribe  a 
continuación, y solo tiene efecto para ese comando. Este comando es @. 

De esta forma, si escribimos @ECHO OFF, no se mostrará el comando ECHO OFF.

@ECHO OFF

ECHO Bienvenido al sistema.

ECHO.

ECHO La fecha y hora actuales son

DATE /T

TIME /T

Código 4 – Línea ECHO OFF sin “eco de comando”

2.2. - Comando REM
El comando REM permite insertar comentarios en un fichero de proceso por lotes. La forma utilizarlo 
consiste en escribir el comentario detrás del comando REM separado con un espacio. Los comentarios 
son de una sola línea.

@ECHO OFF

REM Muestra mensajes de bienvenida y explicación.

ECHO Bienvenido al sistema.

ECHO.

ECHO La fecha y hora actuales son

REM Ahora mostramos la fecha y la hora

DATE /T

TIME /T

Código 5 – Comentarios dentro del proceso por lotes.

2.3. - Comando PAUSE
Para o pausa la ejecución de un proceso por lotes en espera de la intervención del usuario. El intérprete 
continuará ejecutando comandos cuando el usuario pulse una tecla. 

En este punto, el usuario puede terminar con la ejecución del proceso por lotes si lo desea. Para 
ello el usuario puede pulsar la combinación de teclas CTRL+C, el intérprete de comandos pregunta al 
usuario si desea terminar o no con la ejecución del proceso por lotes. Si el usuario contesta no, el 
intérprete continua ejecutando los comandos por el comando siguiente a pause.

Por el contrario, si el usuario contesta Si, el proceso por lotes termina su ejecución en ese punto, sin 
ejecutar ninguno de los comandos que estuvieran después del comando pause.

De esta forma, el comando pause permite solicitar cierta confirmación al usuario para continuar con 
la ejecución del proceso por lotes, pudiendo éste ser detenido antes de llevar a cabo la acción principal 
para la que fue creado. 

En  cualquier  momento  el  usuario  puede  detener  la  ejecución  del  proceso  por  lotes,  como  de 
cualquier comando o aplicación lanzada desde el  intérprete, pulsando la combinación CTRL+C. Sin 
embargo, el uso adecuado del comando pause permite al usuario detener la ejecución del proceso por 
lotes justo antes de que se lleve a cabo alguna acción.

@ECHO OFF

REM Muestra mensajes de bienvenida y explicación.

ECHO Bienvenido al sistema.

ECHO.

ECHO La fecha y hora actuales son

REM Esperamos a que el usuario pulse una tecla.

PAUSE

REM Ahora mostramos la fecha y la hora

DATE /T

TIME /T

Código 6 – Ejemplo con PAUSE: espera a que el usuario pulse una tecla para continuar

2.4. - Parámetros
Uno de los elementos que dota de flexibilidad y potencia a los comandos o aplicaciones en línea de 
comandos es el paso de parámetros. Los parámetros permiten configurar la función de un comando o 
aplicación para adecuarla a lo que se desea.

De la misma forma, los procesos por lotes admiten el paso de parámetros para poder configurar y 
ajustar la tarea que van a realizar. Un proceso por lotes puede recibir entre 1 y 9 parámetros, que se 
pasan como se pasaría a cualquier comando o aplicación.

muestra listado.txt 20

Para obtener los valores pasados por parámetro a un proceso por lotes se utiliza el símbolo “%” 
seguido de un número, que identifica el número del parámetro que se desea obtener. Por ejemplo, si 
quisiera obtener el parámetro pasado en segunda posición o segundo parámetro tendría que poner %2.

Ejercicio Hacer un proceso por lotes, llamado muestra, que muestre por pantalla el primer, segundo y 
tercer parámetro que se le haya pasado. Si lo ejecutáramos con la sentencia anterior debería aparecer 
en pantalla:

Primer parámetro: listado.txt

Segundo parámetro: 20

Tercer parámetro:

Como tercer parámetro no debería mostrar nada, puesto que no se le ha pasado tercer parámetro 
(será una cadena de texto vacía).

Solución:

@ECHO OFF

REM Muestra los tres primeros parámetros pasados.

REM Autor: Javier Pedrero Martín 

REM Fecha: 26/01/2007

ECHO primer parámetro: %1

ECHO segundo parámetro: %2

ECHO tercer parámetro: %3

Código 7 – Solución al ejercicio de paso por parámetro.

Ejercicio ¿Existe el parámetro 0? ¿Para qué vale?

Solución: Si, se trata de la ruta del fichero de proceso por lotes (sin la extensión).

Los parámetros pueden ser utilizados para flexibilizar los procesos por lotes, permitiendo adaptar su 
uso a distintas situaciones. 

Por ejemplo, realiza un proceso por lotes que copie todos los ficheros con extensión C del directorio 
D:\DAI\PLE y todos sus subdirectorios, al directorio D:\Backup. Cuando termines con este proceso por 
lotes, modifícalo para que copie los ficheros con extensión BAT. Si te fijas, lo único que ha cambiado es 
la extensión, aunque te has visto obligado a editar el proceso por lotes. Si esta información viniera por 
parámetro, no habría que volver a cambiar una y otra vez el proceso por lotes cada vez que se quisiera 
realizar una copia de seguridad de un tipo de ficheros distinto.

2.5. - Etiquetas
Las etiquetas se utilizan para nombrar un punto o línea concreta dentro del fichero de procesos por 
lotes.

Su única utilidad es dar nombre a un punto concreto dentro del archivo de procesos por lotes. El 
intérprete de comandos no muestra en pantalla las etiquetas, aunque las tiene en cuenta. 

Para definir una etiqueta, basta con poner “:” seguido del nombre que se le quiera dar a la etiqueta. 
Las etiquetas se deben definir al comienzo de una línea y solo se puede definir una etiqueta en esa 
línea. No se pueden establecer dos etiquetas con el mismo nombre.

Este comando es utilizado en conjunción con el comando GOTO, que se verá a continuación, para 
realizar saltos incondicionales.

2.6. - GOTO
Se trata de un salto incondicional. Permite que el intérprete de comandos salte a una etiqueta definida 
dentro del proceso por lotes. De esta forma, en lugar de seguir ejecutando en secuencia, continuando 
por el  siguiente comando,  tras ejecutar un GOTO el  intérprete de comandos ejecutará el  siguiente 
comando que esté debajo de la etiqueta.

@ECHO OFF

REM etiquetas.bat

REM Autor: Javier Pedrero Martín

REM Fecha: 01/02/2007

REM Ejemplo para el uso de etiquetas

REM Etiqueta de inicio que da nombre a un conjunto de comandos.

:Inicio

ECHO Bienvenido...

ECHO.

REM Salto incondicional a la etiqueta FINAL

GOTO FINAL

REM Se utiliza la etiqueta Mensaje para dar nombre al punto donde comienzan

REM una serie de comandos de mensaje principal

:Mensaje

ECHO Hola esto es una prueba

ECHO en la que se muestran

ECHO distintas cadenas de texto

REM Se utiliza la etiqueta final, para dar nombre al punto del proceso por 
lotes

REM donde se dan mensajes de despedida.

:Final

ECHO Bueno, esto se acaba

ECHO así que me despido... 

ECHO un saludo ;)

ECHO.

Código 8 – Proceso por lotes con etiquetas y saltos.

2.7. - IF
Permite ejecutar un comando solo si se da una determinada condición. Con el modificador NOT se 
puede cambiar  esa condición,  de forma que se ejecutará un determinado comando solo si  NO se 
cumple una condición.

IF <CONDICION> COMANDO

IF NOT <CONDICION> COMANDO

Únicamente se ejecutará un comando en respuesta a la condición, que será el que se escriba detrás 
de la condición en la misma sentencia del IF. 

Las condiciones que se pueden especificar con el comando IF son de tres tipos:

1Comparación de cadenas

IF cadena1==cadena2 comando

IF NOT cadena1==cadena2 comando

2Existencia de archivo

IF EXIST ruta_del_archivo comando

IF NOT EXIST ruta_del_archivo comando

3Resultado de la ejecución de un comando/aplicación.

IF ERRORLEVEL numero comando

IF NOT ERRORLEVEL numero comando

2.7.1. - Comparación de igualdad entre cadenas
La comparación de igualdad entre cadenas se suele utilizar para comprobar si una variable de entorno o 
parámetro del proceso por lotes está inicializada o tiene un valor determinado.

Como ya se ha comentado, cuando el intérprete de comandos encuentra una variable de entorno 
entre símbolos “%” lo que hará será expandir esta cadena, sustituyéndola por su contenido. Lo mismo 
sucede si encuentra un símbolo de % seguido de un número entre 0 y 9, sustituyendo esta cadena por 
el parámetro i-esimo pasado al proceso por lotes.

De  esta  forma,  si  un  parámetro,  por  ejemplo  el  primero,  o  variable  de  entorno,  por  ejemplo 
USUARIO, están vacíos (no inicializados o sin valor), el interprete de comandos los sustituirá por una 
cadena vacía.

ECHO Usuario -%USUARIO%-  Usuario –

ECHO Parámetro -%1-  Parámetro –

Este  comportamiento  hay  que  tenerlo  en  cuenta  cuando  comparemos  cadenas  de  texto  con  el 
contenido de una variable de entorno o de un parámetro, puesto que si dicha variable o parámetro no 
contienen ningún valor la comparación fallará. Para solucionar este problema, bastará con poner entre 
comillas  tanto  la  cadena  de  texto  a  comparar  como  la  viable  o  parámetro,  como  se  muestra  a 
continuación:

IF “%USUARIO%”==”pepito” ECHO Es pepito!

 Este comando funcionará correctamente, puesto que si USUARIO no contiene nada, el comando 
que ejecutará el intérprete tras expandir o sustituir la variable USUARIO será:

IF “”==”pepito” ECHO Es pepito

Si no se pusiera entre comillas:

IF %USUARIO%==pepito ECHO Es pepito!!!

Y el resultado si la variable está vacía sería:

IF ==pepito ECHO pepito

Lo cual produciría un error.

Esto es aplicable a los parámetros del proceso por lotes y también se utiliza con frecuencia para 
comprobar si se ha pasado un valor por parámetro o no.

IF “%1”==”” ECHO No se ha pasado el parámetro 1

La condición anterior  solo se cumplirá cuando %1 esté vacío,  y por tanto “%1” al  expandirlo se 
convierta en “”, cumpliendo así con la igualdad.

2.7.2. - ¿Cómo se pueden ejecutar varios comandos si se cumple una condición?
Aquí se puede utilizar el salto incondicional con el comando GOTO. De esta forma, si se cumple la 
condición tan solo se ejecutará un comando, el comando GOTO, que llevará el flujo de ejecución a la 
etiqueta que se desee.

De esta forma, en la etiqueta destino se puede colocar una secuencia de comandos que lleve a cabo 
la acción deseada en caso de cumplirse la condición.

IF <CONCIDICION> GOTO REALIZA_ACCION

….

:REALIZA_ACCION

REM Se lejecutan los comandos que sean…

2.7.3. - Existencia de archivos 
Con un comando IF  también  se  puede  comprobar  si  existe  un  archivo  (fichero  o  directorio).  Esta 
comprobación puede ser interesante para llevar a cabo una tarea concreta dentro de nuestro proceso 
por lotes.
IF EXIST ruta_del_archivo comando
IF NOT EXIST ruta_del_archivo comando

2.8. - Call
Se utiliza para invocar a un proceso por lotes dentro de otro proceso por lotes.  Si  no se utiliza el 
comando call para llamar a un segundo proceso por lotes desde un proceso por lotes, cuando termine el 
segundo proceso por  lotes,  el  primero terminará también.  Esto hará  que no se siga  ejecutando el 
proceso por lotes que invocó la llamada.

Para  solucionar  este  inconveniente  se  puede  utilizar  el  comando  call,  seguido  del  nombre  del 
proceso por lotes a invocar y de los parámetros que se estimen oportunos.

Call creaRuta C:\dats

2.9. - For
El comando For permite repetir  un comando con un conjunto de ficheros.  Se permite que en cada 
ejecución del comando, una variable tome el valor del nombre del fichero i-esimo.

La sintaxis es la siguiente:

FOR %%<Variable> IN (<conjunto_ficheros>) DO <comando>

El nombre de la variable debe ser de una sola letra y,  además, deberá estar precedido por dos 
símbolos “%”3. Un pequeño ejemplo del uso de este comando en un proceso por lotes puede ser el 
siguiente:

3 Esto es así solo en los procesos por lotes. Si se utiliza este comando desde consola de comandos bastará con que 
vaya precedido por un solo símbolo “%”

@ECHO OFF

REM Se va a mostrar el nombre de todos los ficheros con extensión bat y CMD 
dentro del

REM directorio actual

FOR %%I IN (*.BAT *.CMD) DO ECHO Fichero: %%I

Código 9 – Muestra ficheros con extensión BAT o CMD.
El comando ECHO Fichero: %%I se repetirá tantas veces como ficheros BAT y CMD haya en el 

directorio actual. Esto es así debido a que el conjunto de ficheros se ha definido de esta forma, es decir, 
utilizando el comodín * seguido de .BAT y .CMD respectivamente.

… %%I IN (*.BAT *.CMD) DO …

En cada repetición, la variable %%I tomará como valor el nombre de cada uno de estos ficheros. Por 
eso se muestra el nombre de cada fichero en pantalla al utilizar el comando ECHO.

… ECHO Fichero: %%I

La siguiente línea de un proceso por lotes copiará los ficheros con extensión MP3 que comiencen por 
“master” al directorio D:\Musica\

FOR %%I IN (master*.MP3) copy %%I D:\Musica

2.10. -Shift
La traducción  de la  palabra  Shift  de  inglés  a  español  es  “desplazar”,  y  esto  es  lo  que hace  este 
comando: “desplazar los parámetros de entrada de izquierda a derecha”. 

Cada vez que se ejecuta el comando Shift dentro de un proceso por lotes, el intérprete de comandos 
desplaza los parámetros pasados al proceso por lotes una posición a la izquierda. De esta forma, el 
parámetro 1 (%1) pierde su valor anterior y pasa a tomar el del parámetro 2 (%2) y así sucesivamentede 
forma  escalada.  El  parámetro  0  también  se  ve  alterado  por  este  comando,  tomando  el  valor  del 
parámetro 1.

El siguiente ejemplo, ilustra lo que sucede con el comando Shift y como podemos comprobar los 
parámetros que se pasan a un proceso por lotes uno tras otro.

@ECHO OFF

REM Vamos a recorrer los parámetros pasados al proceso por lotes uno a uno.

REM Bloque principal del proceso, que se repetirá hasta que no haya parámetros

:PPAL

IF "%1"=="" GOTO FIN

ECHO Parámetro: %1

SHIFT

GOTO PPAL

REM Termina el proceso por lotes

:FIN

Código 10 – Ejemplo del comando Shift: proceso por lotes parámetros.bat

Si este proceso por lotes se llamara parámetros.bat y se ejecutase así:

parametros uno dos tres cuatro cinco

Daría la siguiente salida:

Parámetro: uno

Parámetro: dos

Parámetro: tres

Parámetro: cuatro

Parámetro: cinco

Javier Pedrero Martín

